
Önsöz 
 

Millet olarak yılbaşlarında, bayramlarda tebrik kartı, hediye göndermeyi severiz. Bu sene ben bir 

değişiklik yapmak istedim, elektronik bir Excel kitabı hazırlayarak sizinle paylaşmaya karar verdim. Bu 

hediyeyi siz de arkadaşlarınıza gönderebilirsiniz. Sevgi ve bilgi paylaştıkça çoğalır. 

Haklısınız, “VLOOKUP Hakkında Herşey” iddialı bir isim, belki de “VLOOKUP Hakkında Benim Bildiğim 

Herşey” demem daha doğru olurdu. Eksik bulduğunuz bir konu varsa lütfen yazın. Olumlu, olumsuz 

tüm görüşlerinizi excel@vegatr.com adresine gönderebilirsiniz. 

 

 

Aralık 2014, Atlanta 

Okan Emir  

  

mailto:excel@vegatr.com

e-Kitap Hakkında  
 

Bu e-kitapcık, İleri Excel kullanıcılarını diğerlerinden ayıran en temel konu olan VLOOKUP ve benzeri 

fonksiyonlara ayrılmıştır. Excel kullanıcıları ikiye ayrılır, VLOOKUP yapabilenler ve VLOOKUP 

yapamayanlar. VLOOKUP yapabilenler önemli bir eşiği atlamış, Excel’i fethetmiş sayılır.  

Temel VLOOKUP kullanımı ile başlayıp, MATCH ile istediğimiz kolondan VLOOKUP yapmayı, iki veya 

daha fazla kolon üzerinde VLOOKUP yöntemini, INDEX - MATCH ile soldaki değere ulaşma yöntemini, 

VLOOKUP ile birlikte kullanılan diğer teknikleri göreceğiz.  

Kitapta yer alan örnekleri www.vegatr.com/VLOOKUP adresinden bilgisayarınıza indirip, ihtiyacınız 

olduğunda kolayca bakabilirsiniz. Dilerseniz sadece o anda ulaşmak istediğiniz örnekleri inceleyin, 

dilerseniz tamamını okuyun.  

Bu kitap VLOOKUP konusunda benim bildiğim herşeyi içeriyor, umarım beğenirsiniz, umarım işinize 

yarar. 

 

 

 

 

 

 

 

 

 

 

Örnekleri www.vegatr.com/VLOOKUP adresinden indirebilir, arkadaşlarınızla paylaşabilirsiniz. 

 

  

http://www.vegatr.com/VLOOKUP
http://www.vegatr.com/VLOOKUP

Giriş  
 

Microsoft Excel 2015 yılında 30. yaşını kutluyor. Dünyada yüz milyonlarca insan Excel kullanıyor, iş 

dünyasında, finans dünyasında, mühendislikte Excel standardı oluşmuş durumda. İş arayanların, 

işinde ilerlemek isteyenlerin Excel’de hazırlanmış modelleri hızlı anlaması, düzeltmesi, geliştirmesi 

bekleniyor. 

Excel, hesap tablosu (Spread Sheet) işlemleri için tasarlanırken, büyük veriler için Microsoft Access ve 

SQL sistemleri tasarlanmış. İdeal olarak 65,000 satırdan az veriler Excel’de işlenmeli, bir kaç milyon 

satıra kadar veriler MS Access’te, daha büyük veriler ise SQL sistemlerde bilgi işlem uzmanlarınca 

işlenmelidir.  

Pratikte ise bir finansçıdan herşeyi yapması, onbinlerce satırdan oluşan bir tabloyu onbinlerce satırlık 

tablolarla birleştirmesi, birinde olan, diğerinde olmayan kayıtları yakalaması, çift kayıtları silmesi, 

bilgi işlemden destek almadan zorlu raporlar üretmesi bekleniyor. Excel 2007’nin 1,048,000 satırlık 

yeni limiti, büyük müşteri listelerini, uzun tabloları Excel’de tutmaya imkan veriyor. 

Excel’de veri tabanı işlemleri yapmaya en uygun fonksiyon VLOOKUP. VLOOKUP ile iki tabloyu 

birleştirmek, bir listede olup öbür listede olmayan kayıtları bulmak mümkündür. İşin uzmanıysanız (ki 

bu kitabı okuyunca olacaksınız) daha fazlasını da yapabilirsiniz. 

Temel VLOOKUP ile başlayıp, iki boyutlu VLOOKUP ile devam edeceğiz. Performans problemi 

yaşayanlar için hızlı VLOOKUP, soldaki kolondan LOOKUP, üç ve daha fazla boyutlu LOOKUP örnekleri 

göreceğiz. 

 

 

Örneklerin tamamına www.vegatr.com/VLOOKUP adresinden ulaşabilirsiniz. 

 

 

 

 

 

 

  

http://www.vegatr.com/VLOOKUP

01 Basit VLOOKUP 
 

Birinci örneğimiz “01 Basit VLOOKUP.xlsx” dosyası (www.vegatr.com/VLOOKUP). 

 

VLOOKUP’ın en temel kullanımı =VLOOKUP( Aradığımız değer, Taranan liste, Kaçıncı kolon, 0 ) 
şeklindedir.  

=VLOOKUP( F6, A2:D11, 2, 0 ) 

=DÜŞEYARA( F6; A2:D11; 2; 0 ) 

Yukarıdaki örnekte aradığımız değer F6 hücresinde yer alan plaka numarasıdır. F6 hücresindeki plaka 

numarası değiştikçe G6, H6 ve I6 hücrelerindeki değerler de değişir. 

Taranan liste A2:D11 alanının en soldaki kolonu yani A2:A11 alanındaki plakalardır. 

İkinci kolon seçilerek B2:B11 alanındaki bilgilerin seçilmesi sağlanmıştır. G6 hücresi için 2. kolon, H6 

için 3. kolon, I6 için 4. kolondaki bilginin gelmesi sağlanmıştır. 

Dördüncü parametre 0 (False / Yanlış) “Tam eşleşme” anlamına gelir. İş hayatında şahit olduğum 

VLOOKUP kullanımlarının %99.9’u “Tam eşleşme” gerektiren türden olmuştur.  

1 (True / Doğru ) “Yaklaşık eşleşme” kullanımı listeden belli aralığa karşılık değer atamak için kullanılır 

(otopark tarifesinde 2 - 4 saat arası ücretlendirmede olduğu gibi). 

  

http://www.vegatr.com/VLOOKUP

VLOOKUP( Aradığımız değer, Taranan liste, Kaçıncı kolon, Arama Türü ) 

VLOOKUP Vertical Lookup (Dikey Bakmak/Aramak) kelimelerinden türetilmiştir. Taranan listede 

aranan değeri bulduğu ilk satırda durur. İstenilen kolonda yer alan değeri getirir. Türkçe Excel’deki 

karşılığı DÜŞEYARA şeklindedir. 

 

Aradığımız değer : Numerik veya metin bir değer olabilir.  

Taranan liste : Verilerin bulunduğu alandır, aranan değer ilk kolonda bulunur. “Tarama” kelimesinin 

amacı VLOOKUP fonskiyonunun binlerce satırlık listelerde, her hücreyi tek tek taradığını belirtmek 

içindir. 

Kaçıncı kolon : Listeden hangi kolon bilgisinin getirileceğini belirtir. Aradığımız değerin sağında 

bulunan kolonlar seçilebilir. Sol tarafta bulunan kolonları seçmek için INDEX+MATCH ikilisi veya 

vektörel LOOKUP kullanılmalıdır. 

Arama türü : Dördüncü parametre 0 - FALSE (tam eşitlik) veya 1 – TRUE (yaklaşık değer) değeri 

alabilir. Excel’de kullanılan VLOOKUP formüllerinin çok büyük bir kısmında 0 - FALSE (tam eşitlik) 

parametresi kullanılmaktadır. Bu durumda liste baştan sona kadar taranır, aradığımız değere uyan bir 

satır bulunarak o satırdan istediğimiz kolonun değeri getirilir. 

Dördüncü parametrenin 1 – TRUE (yaklaşık değer) olarak kullanılabilmesi için olmazsa olmaz bir koşul 

vardır; aranan liste artan sıralı olmalıdır. Aradığımız değere eşit olmasa da, daha küçük 1 en yakın 

değere sahip satırdaki bilgi getirilir.  

 

 

  

                                                           
1 Daha büyük en yakın değer getirilmesi gerekiyorsa, INDEX-MATCH ikilisi kullanılabilir. 

02 VLOOKUP Genel Kullanım 
 

 

VLOOKUP formülü genellikle iki listeyi birleştirmek için kullanılır. Sabit bilgilerin yer aldığı bir liste ile 

sık değişen bilgileri (araçların kilometresi gibi) bir araya getirmek için VLOOKUP formülleri kullanılır.  

Eğer her iki liste aynı sayıda satır içeriyorsa ve aynı sırayla geliyorsa sorun yok, Copy-Paste ile sağdaki 

listeden bilgileri alır, soldaki listeye eklersiniz. Gerçek hayat şartları ne yazık ki bu durumdan çok 

uzaktır. Listeler farklı sırada gelir, bazı satırlar eksiktir, fazladan satırlar vardır, v.s. Bu durumda 

verileri birleştirmek için Excel’de VLOOKUP yardımınıza koşar. 

Yukarıdaki örnekte Kolon 1, 2, 3, 4 soldaki listeden gelirken 5. kolon sağdaki listeden gelmektedir. 6. 

kolon ise 4. ve 5. kolonlar kullanılarak hesaplanmaktadır. 

=VLOOKUP( A2 , I:J , 2 , 0 )  

=DÜŞEYARA( A2 ; I:J ; 2 ; 0 )  

Yukarıdaki formül E2 hücresine yazılarak aşağıya doğru (E2:E11) kopyalanır ve iki liste birleştirilir. 

Bu örnekte taranan alan olarak I2:J14 yerine I:J kolonları seçilmiştir. Eğer I ve J kolonlarının alt 

satırlarında başka veriler varsa $I$2:$J$14 şeklinde kullanmanız veya ad tanımlamanız (named range) 

doğru olacaktır. Benim I:J şeklinde kullanmam biraz tembellik, biraz pratikten kaynaklanıyor   

Ben Nerede Yanlış Yaptım ? 
 

VLOOKUP formülü yazdınız, aşağıya doğru kopyaladınız ve hatalarla karşılaştınız. Duruma göre farklı 

nedenleri inceleyelim : 

1. Aralarda az sayıda #N/A (#YOK) hatası var 

Bu normal bir durumdur. Taranan listeye, aradığınız değeri eklerseniz sorun çözülecektir. 

Yukarıdaki örnekte “34 DE 9684” plakalı araç soldaki listede varken, sağdaki listede bulunmadığı 

için #N/A (#YOK) hatası oluşmuştur. 

#N/A hatalarını hiç görmek istemiyorsanız IFERROR (EĞERHATA) fonksiyonu işinizi görür. 
IFERROR( VLOOKUP( .... ), “”) Hata durumunda hücreye boş bir metin getirir. 
IFERROR( VLOOKUP( .... ), 0) Hata durumunda hücreye sıfır sayısı getirir. 

2. Başlarda az sayıda #N/A (#YOK) hatası var, sonlara doğru tamamı hataya dönüyor 

Bu durum taranan listenin adresini $ işareti olmadan yazmaktan kaynaklanır. $I$2:$J$14 

yazmaya üşendiniz, I2:J14 yazdınız ve ilk satırda formül çalıştı öyle değil mi? Tamam dediniz ve 

hücreyi alttaki satırlara kopyaladınız, keşke önce kullanım kılavuzunu okusaydınız  

E11 hücresine gelince formülünüz =VLOOKUP(A11,I11:J22,2,0) şekline dönüştü. 

 

3. Tüm satırlarda #N/A (#YOK) hatası var 

Verileri dikkatle gözden geçirmeniz gerekiyor. Aynı gibi gördüğünüz “34 DE 9683” ile 

“34 DE   9683” farklı metinlerdir. Baştaki, sondaki boşluklar, aradaki boşluk sayısının farklı olması  

bu hatalara yol açabilir. TRIM (KIRP) fonksiyonu işinize yarayabilir. 

4. Hiç #N/A (#YOK) hatası yok ama doğru sonuçlar gelmemiş 

Dördüncü parametre hep 0 olmalı demiştik, hatırladınız mı? Dördüncü parametreyi boş 

bırakmak veya 1 (TRUE) yazmak, VLOOKUP’ın sıralı listede arama yapmasına yol açar, #N/A 

hatası almazsınız ama sonuçlar hatalı gelir. 

 

5. Metin içeren hücreler tamam, sayı olanlarda #N/A (#YOK) hatası var 

Excel hücrelerine sayıları metin olarak saklamak mümkündür. Hücrenin sol üst köşesinde küçük 

yeşil bir uyarı üçgeni oluşur ( gibi) ve “Metin olarak saklanan sayı” mesajı görünür.  

0 ile başlayan hesapları girebilmek için bu yöntem özellikle tercih edilmiş olabilir. Sayının başına 

tek tırnak ( ' ) ekleyerek veya hücre biçimlendirmesini “Metin” (Format Cell / “Text”) yaparak 

sayıları metin olarak saklamak mümkündür.  

E kolonu hücre biçimlendirmesi “Metin” (Format Cells / “Text”) yapılmıştır. A kolonuna ise işlem 

yapılmadan 010, 020 gibi rakamlar girilmiştir. E kolonu sola dayalıdır (metin), A kolonu ise sağa 

dayalıdır. 

Eğer bazı sayılar metin, bazıları sayı olarak kaydedilmiş ise başınız dertte demektir. Ya tamamını 

sayıya çevirin ya da tamamını metin yapın. VLOOKUP aynı tür olanları eşleştirebilir. 

 Listeyi seçip, metin olarak kaydedilmiş bir hücreye geldiğinizde Convert to Number (Sayıya 
Çevir) seçeneği metin olarak saklanmış tüm değerleri sayıya çevirir.  

 

6. Aradığım değer taranan listede birden fazla var, hep ilki geliyor. 

VLOOKUP, tanımından dolayı hep ilk değeri getirir.  

 Eğer ilk değil de son değeri istiyorsanız, taranan listeyi ters sıraya dizin. 

 Eğer ortalamayı istiyorsanız AVERAGE formülünü kullanın. 

 Eğer toplam istiyorsanız SUMIFS formülünü kullanın. 

 

7. Tarih değerini almak istiyorum, garip sayılar (41768 gibi) geliyor. 

Tarih değerlerini Format Cells (Hücre Biçimlendirme) ile düzenlerseniz sorun çözülür. 

03 VLOOKUP Data Validation 
 

 

Excel’in gerçek gücü birden fazla fonksiyonu, birden fazla yöntemi bir arada kullanmaya başlayınca 

ortaya çıkar.  

Bu örnekte D6 hücresinde “Data Validation” (Veri doğrulama) kullanılmıştır. D6 hücresini seçtiğinizde 

sağ tarafında bir üçgen kutucuk belirir ve listeden değer seçmenizi sağlar. Yaptığınız Excel tablo daha 

etkileyici görünür. 

 

nrPlakalar =Veri!$A$2:$A$11 



12 
 

 

Veri isimli çalışma sayfasında iki “Ad” tanımlanmıştır, nrListe, nrPlakalar. Formüllerde 

Veri!$A$2:$A$11 şeklinde kullanmak yerine nrPlakalar, Veri!$A$2:$D$11 şeklinde kullanmak yerine 

nrListe adları (Named Range) kullanmıştır. 

Veri!$A$2:$D$11 şeklinde tanımlamam gereken adresi çoğunlukla unutup Veri!A2:D11 şeklinde 

yazarım. Formülü sağa ve aşağıya çektiğim zaman hatayı farkedip düzeltirim. Ad kullanımı bizi 

VLOOKUP formülü içinde mutlak adres kullanma zorluğundan kurtardığı gibi formüllerin daha rahat 

okunmasını, anlaşılmasını da sağlar. 

04 Tablodan VLOOKUP-MATCH 
 

 

VLOOKUP fonksiyonuna biraz takviye yaparak, tablonun istenilen satırı, istenilen kolonundaki değeri 

getirebiliriz. Bunun için D15 hücresinde MATCH fonksiyonunu kullanarak “Nisan” ayının hangi 

kolonda yer aldığını buluruz. 

D15 =MATCH( B15 , B1:M1 , 0 )  +  1 

 =KAÇINCI( B15 ; B1:M1 ; 0 )  +  1 

E15 =VLOOKUP( A15 , A2:M11 , D15 , 0 ) 

 =DÜŞEYARA( A15 ; A2:M11 ; D15 ; 0 ) 

D15 hücresinde bulduğumuz kolon numarasını E15 hücresindeki formüle yerleştirdiğimizde 

aradığımız plakalı aracın, aradığımız aydaki kilometre değerini bulabiliriz. 

MATCH - KAÇINCI 

MATCH fonksiyonu, aranan değerin kaçıncı sırada olduğunu verir. Genellikle INDEX fonksiyonuyla 

birlikte kullanılır. 

=MATCH( Aradığımız değer, Aranan vektör, Arama Türü ) 

 

Aradığımız değer : Numerik veya metin bir değer olabilir.  

Aranan vektör : Aranan değerin bulunduğu bir kolon veya bir satırdır (sıralı olmak zorundadır). 

Arama türü : -1, 0 ve 1 değerlerini alabilir. 0 tam eşitlik anlamına gelir ve aradığımız değere tam 

olarak eşit değeri bulana kadar sırayla arama yapılır, bulunamazsa #N/A hata değeri döner. -1 ve 1 

daha büyük ve daha küçük değeri döndürür. -1 kullanabilmek için aranan listenin azalan sırada, 1 

kullanabilmek için aranan listenin artan sırada olması gerekir. Eğer çok uzun listelerle 

uğraşmıyorsanız 0 kullanmak tam eşitliği sağladığı için garantili olacaktır. 

05 VLOOKUP Wildcard  
 

 

VLOOKUP fonksiyonunda * ? gibi karakterleri kullanabileceğinizi biliyor muydunuz? 

*ZZ Sonu “ZZ” ile biten ilk kayıdı bulur. 

Örnek XXXZZ, abczz, t000zZ, ZZ 

AA* “AA” ile başlayan ilk kayıdı bulur. 

Örnek AABBB, AA, Aaxyz, aa0099 

AA*ZZ “AA” ile başlayan ve sonu “ZZ” ile biten ilk kayıdı bulur.   

Örnek AAZZ, aaxxxzz, Aaxyzz, AAZZ 

A?Z “A” ile başlayan, herhangi tek bir karakter ile devam eden, “Z” ile biten ilk kayıdı bulur. 

Örnek AOZ, axz, A-Z 

A???Z “A” ile başlayan, üç karakter ile devam eden, “Z” ile biten ilk kayıdı bulur. 

Örnek AOOOZ, axyzz, A123Z 

06 İki Kolona VLOOKUP 1 
 

 

İki kolondaki bilgiyi VLOOKUP ile aramanın en kestirme yolu “Yardımcı kolon” kullanmaktır. Yardımcı 

kolonda “Kod” ve “Boy” bilgileri birleştirilir, VLOOKUP ile arama işlemi bu kolon üzerinde yapılır. 

A2 =B2 & "|" & C2 

F9 hücresinde “Kod” bilgimiz (Data Validation ile) 

G9 hücresinde “Boy” bilgimiz (Data Validation ile) 

H9 =F9 & "|" & G9 

I9 =VLOOKUP ( H9 , A:D , 4 , 0 ) 

I9 =DÜŞEYARA( H9 ; A:D ; 4 ; 0 ) 

“Kod” ve “Boy” bilgilerini birleştirirken araya “|” eklememizin sebebine gelince … 

Farzedin ki “Ad” ve “Soyad” kolonlarını birleştirerek arama yapıyoruz ve listemiz aşağıdaki gibi: 

 

Birleştiğinde aynı sonucu verecek isimleri özellikle seçtim. “Ali Canyurt” ile “Alican Yurt” eğer araya 

ayraç koymazsak Excel için aynı sonucu verecek ve sonuçlar hatalı olacaktır. I kolonundaki sonuçların 

yarısı hatalıdır. 

Doğru sonuca ulaşmak için “ . , / + | ” gibi ayraçlardan herhangi birini kullanabiliriz. Benim favorim | 

işaretidir. 

İkiden fazla kolonu da bu yöntemle birleştirmek için formülünüz : 

A2 =B2 & "|" & C2 & "|" & D2 & "|" & E2 ….. 

şeklinde olacaktır. 

07 İki Kolona VLOOKUP 2 Ctrl-Shft-Enter 
 

 

Eğer yardımcı kolon kullanmadan iki kolona VLOOKUP yapabiliyorsanız size Excel Gurusu diyebiliriz. 

Yapacağınız tek şey G9 hücresine  

=VLOOKUP ( E9 & "|" & F9 , CHOOSE({1,2} , $A:A & "|" &$B:B , $C:C ) , 2 , 0)  

=DÜŞEYARA ( E9 & "|" & F9 ; ELEMAN( {1,2} ; $A:A & "|" &$B:B ; $C:C ) ; 2 ; 0)  

Formülünü yazıp “Ctrl + Shift + Enter” tuşlarına birlikte basmaktır. Böylece formülünüz 

{=VLOOKUP ( E9 & "|" & F9 , CHOOSE({1,2} , $A:A & "|" &$B:B , $C:C ) , 2 , 0)} 

Şeklinde görünür. Sakın ola “{“ “}” işaretlerini klavyeden yazarak eklemeye kalkmayın, sonuç 

alamazsınız. “Ctrl + Shift + Enter” Array Formula (Dizi formülü) oluşturmak için kullanılan yöntemdir. 

Sadece bu üç tuşa birlikte basmanız gerektiğini bilin yeter. 

İkiden fazla kolonu da bu yöntemle birleştirmek için formülünüz : 

=VLOOKUP ( S1 & "|" & S2 & "|" & S3 & "|" & S4 …. , CHOOSE({1,2} , Kolon 1 & "|" & Kolon 2 

& "|" & Kolon 3 & "|" & Kolon 4 …  , Sonuç Kolonu ) , 2 , 0)  

şeklinde olacaktır. 

Örnek : 

{= VLOOKUP ( F9 & "|" & G9 & "|" & H9 , CHOOSE({1,2}, $A:A & "|" &$B:B & "|" &$C:C , $D:D ),2,0) } 

  Uyarı : Ctrl-Shift-Enter formülleri yavaş çalıştıkları için, çok gerekmedikçe kullanılmamalıdır.|        

08 VLOOKUP Yaklaşık Eşleşme 
 

 

VLOOKUP fonksiyonunu açıklarken, dördüncü parametrenin büyük çoğunlukla 0 yapılması 

gerektiğinden bahsetmiştik. Bire bir eşleşme durumlarında 0 “Tam eşleşme” kullanırken, bir aralığa 

uygun değer getirebilmek için 1 “Yaklaşık eşleşme” kullanmamız gerekir. 

Yukarıdaki örneğin sağında G1:I9 alanında ciroya göre iskonto tarifesi yer almakta.  

Solda C kolonundaki ciroya uyan iskonto oranlarını D kolonunda görmek istiyoruz. 

Akla gelen ilk yöntem içiçe IF kullanmak olurdu. Oranlar değiştikçe Excel sayfalarına dağılmış 

formülleri bulup değiştirmek çok da hoş bir yöntem sayılmaz.  

D kolonundaki formülümüz : 

=VLOOKUP( C2 ,  $G:$I , 3 , 1 )  

=DÜŞEYARA( C2 ; $G:$I ; 3 ; 1 ) 

şeklindedir. Dördüncü parametrenin 1 olması, tam eşleşme yerine yaklaşık eşleşme kullanarak  

iskonto tarifesini uygulamamızı sağlar. 

09 Hızlı VLOOKUP 
 

 

“Hızlı VLOOKUP da ne demek, bilgisayarlar yeterince hızlı değil mi zaten?” diye soruyorsanız, hayatın 

sillesini henüz yememişsiniz demektir. VLOOKUP fonksiyonu 10,000 satırlık bir listede aranan değeri 

yaklaşık olarak saniyenin binde birinde bulur. “Bu yeterince hızlı” diyorsanız bir de şunu dinleyin : 

10,000 satırlık iki listeyi birleştiriyorsunuz ve VLOOKUP kullandığınız 3 kolon var. 30,000 tane 

VLOOKUP çalıştıracaksınız ve bu işlem 30 saniyenizi alacak demektir.  

Excel dosyayı her açtığınızda, her kaydettiğinizde, hücreleri değiştirdiğinizde karşınızda 30 saniyelik 

bir engel var demektir. Köprüyü geçmek gibi dertli bir iş. “Bilgisayarımı hızlandırırım, RAM’imi 

artırırım” demeyin, bunlar sizi kurtaramaz.  

VLOOKUP’ın hızlı çalışabilmesi için “Taranan liste”nin artan sıraya dizilmesi birinci şartımız. Data / 

Sort (Veri / Sıraya diz) işlemi ile bunu kolayca yapabilirsiniz. 

VLOOKUP’ın hızlı çalışabilmesi için ikinci şartımız aşağıdaki formül yapısını kullanmanız : 

=IF( VLOOKUP( Aranan değer, Taranan liste, 1, 1)= Aranan değer, VLOOKUP (Aranan değer, Taranan 

liste , Kaçıncı Kolon, 1), #N/A) 

=EĞER( DÜŞEYARA(Aranan değer; Taranan liste ; 1; 1)= Aranan değer; DÜŞEYARA(Aranan değer; 

Taranan liste ; Kaçıncı kolon; 1); #YOK) 

Yaklaşık eşleşme kullanmak (dördüncü parametrenin 1 olması), arama işleminin sıralı bir liste 

üzerinde yapılacağını belirtir.  Sıralı listede arama bilgisayarın en hızlı yaptığı işlemdir. Sıralı olmayan 

10,000 satırlık bir listede, aranan değerin bulunması için 10,000 satırın tamamını taramak gerekirken, 

sıralı bir listede ( 10,000 < 16,384 = 2^14), 14 aramada aranan kayıt bulunur.  

Formülün başındaki “VLOOKUP( Aranan değer, Taranan liste, 1, 1)= Aranan değer” kısmı ile 

aradığımız değerin listede olup olmadığını kontrol ederiz, eğer varsa ikinci VLOOKUP ile almak 

istediğimiz kolondaki değeri getiririz, yoksa #N/A değerini getiririz. 

Bu yöntemi 2013 basımlı “İleri Excel” kitabımda anlattıktan sonra, bir banka çalışanı, VLOOKUP 

formülleriyle dolu olan ve hesaplanması 10 saatten fazla süren Excel dosyasının bu yöntemle 5 

dakikada bittiğini belirtmişti. Bu mesajın beni ne kadar mutlu ettiğini anlatamam. 

10 Listeden ilk 5 VLOOKUP 
 

 

Raporlarda ürünleri alt alta, her ürünün ilk n değerini yan yana veriyorsanız bu örnek tam size göre.  

Önce veri sayfamızı inceleyelim : 

 

A kolonu “Yardımcı Kolon” olarak kullanılmakta, aşağıdaki formül ile ürünlerin kaçıncı kez tekrar 

ettiği bulunmaktadır.  

=B2 & “|” & COUNTIF( $B$2:B2 , B2 ) 

=B2 & “|” & EĞERSAY( $B$2:B2 ; B2 ) 

B:F kolonlarında Ürün adı, Tarih, Adet, Fiyat ve Tutar bulunmaktadır. 

Rapor sayfasını adım adım oluşturalım. 

 J1 hücresine “Secim” adını tanımlayın. 

 

 K1:K3 alanına Adet, Fiyat, Tutar yazdıktan sonra J2 hücresine  

=INDEX ( K1:K3 , J1 ) yazın.  

 

J1 hücresinin 1, 2, 3 değerlerini alması için ListBox ekleyeceğiz. Developer tab / Insert / Form controls 

/ Listbox (Geliştirici sekmesi / Ekle / Form denetimleri / Listbox) seçip aşağıda görülen ayarları yapın. 

 

  

 Raporun başlıklarını hazırlamak için birinci satıra Sıra No, 1, 2, 3, 4, 5 yerleştirin. 

 İkinci satıra “Ürün” yazdıktan sonra, “Tutar, Adet, Fiyat” seçimlerini görebilmek için C2’ye 

= $J$2 & “ “ & C1 yazın. Formülü C2:G2 alanına kopyalayın. 

 

Ürün listesini B kolonuna kopyalayın. 

C3 hücresine aşağıdaki formülü yazıp D, E, F, G kolonlarına kopyalayın (C3:G49 alanına). 

=IFERROR ( VLOOKUP ( $B3 & "|" & C$1 , Veri!$A:$F , 3 + Secim ,  0 ) ,  "" ) 

=EĞERHATA( DÜŞEYARA($B3 & "|" & C$1 ; Veri!$A:$F ; 3 + Secim ; 0 ) ; "" ) 

 

Rapor sayfasında sağ tarafta yer alan liste kutusu (ListBox), “Adet, Fiyat, Tutar” kolonlarını seçmemizi 

sağlar.  

VLOOKUP Ötesi 
 

VLOOKUP fonksiyonunu yoğun kullananlar, çoğu kez diğer adresleme fonksiyonlarını da kullanmak 

zorunda kalırlar. Bunlardan en meşhuru INDEX ve MATCH (İNDİS-KAÇINCI) ikilisidir.  

Bu ikilinin dışında, HLOOKUP, LOOKUP, OFFSET, INDIRECT, CHOOSE, ROW, COLUMN  (YATAYARA, 

ARA, KAYDIR, DOLAYLI, ELEMAN, SATIR, SÜTUN) fonksiyonları da yerine göre kullanılabilir. 

Şimdi diğer fonksiyonların kullanımını hızlıca inceleyelim : 

INDEX ( liste, satır, [kolon] ) : Listeden istenilen satır ve kolondaki değeri verir. 

MATCH ( aranan değer, aranan vektör, arama türü ) : Aranan değerin kaçıncı sırada olduğunu verir. 

HLOOKUP ( aranan değer, liste, satır numarası, yaklaşık_tam değer ) : Bir listenin en üst satırına bakar 

ve belirtilen hücrenin değerini verir. 

LOOKUP ( aranan değer, aranan dizi ) : Bir dizideki değerleri arar. 

LOOKUP( aranan değer, aranan vektör, sonuç vektör ) : Bir vektördeki değerleri arar. 

OFFSET ( referans, satır kaydır, kolon kaydır ) : Verilen bir hücreden istenilen satır ve kolon uzaklıktaki 

hücreyi verir. 

OFFSET( referans, satır kaydır, kolon kaydır, satır sayısı, kolon sayısı ) : Verilen bir hücreden istenilen 

satır ve kolon uzaklıkta bir alan getirir. 

INDIRECT (Referans) : Metin değeriyle belirtilen bir hücreyi verir. 

INDIRECT( Referans, Referans_Türü ) 

Referans:  A1 veya R1C1 tarzında hücre adresi verilir. 

Referans Türü :  A1 tarzı için 1 / TRUE, R1C1 tarzı için 0 / FALSE değeri kullanılır. 

CHOOSE : Değerler listesinden bir index ile değer seçer. 

CHOOSE( İndex değer, değer 1, değer 2, değer 3, .... ) 

ROW( referans ) : Bir başvurunun satır numarasını verir. 

COLUMN( referans ) : Bir başvurunun kolon numarasını verir.  



26 
 

11 HLOOKUP 
 

 

VLOOKUP’ın kardeşi HLOOKUP kolon yerine satırlarla çalışır. Yapısı VLOOKUP ile aynıdır. 

HLOOKUP( Aradığımız değer, Taranan liste, Kaçıncı satır, Arama Türü ) 

HLOOKUP Horizontal Lookup (Yatay Bakmak/Aramak) kelimelerinden türetilmiştir. Taranan listede 

aranan değeri bulduğu ilk kolonda durur. İstenilen satırda yer alan değeri getirir. Türkçe Excel’deki 

karşılığı YATAYARA şeklindedir. 

 

Aradığımız değer : Numerik veya metin bir değer olabilir.  

Taranan liste : Verilerin bulunduğu alandır, aranan değer ilk satırda bulunur.  

Kaçıncı satır : Listeden hangi satır bilgisinin getirileceğini belirtir. Aradığımız değerin altında bulunan 

satırlar seçilebilir. Üst satırları seçmek için INDEX+MATCH ikilisi veya vektörel LOOKUP 

kullanılmalıdır. 

Arama türü : Dördüncü parametre 0 - FALSE (tam eşitlik) veya 1 – TRUE (yaklaşık değer) değeri 

alabilir.  

12 LOOKUP ile Soldaki Değer 
 

Excel’in 1985’deki ilk versiyonunda (Lotus 1-2-3 uyumlu) LOOKUP fonksiyonu vardı, VLOOKUP ve 

HLOOKUP daha sonra Excel’e eklendi. Yani LOOKUP, VLOOKUP’ın babasıdır diyebiliriz. 

VLOOKUP o kadar başarılı oldu ki, herkes LOOKUP’ı unuttu. Eski günleri yad etmek isterseniz LOOKUP  

hala işinizi görecektir, yeter ki tarama yaptığınız liste artan sıralı olsun. 

LOOKUP (ARA) fonksiyonunu iki formda kullanabilirsiniz, dizi formu ve vektör formu. 

 

Dizi formu (Array form) : 

LOOKUP(Aradığımız değer, Taranan dizi ) 

Aradığımız değer : Numerik veya metin bir değer olabilir.  

Taranan dizi : Aranan değer dizinin ilk kolonunda, sonuç değer dizinin son kolonundadır. Aranan 

kolon artan sırada olmak zorundadır. 

 

Dizi formu VLOOKUP ile benzer şekilde çalışır, aradığımız değer dizinin sol kolonunda aranır (A 

kolonu), en sağdaki kolondan değer getirilir (D kolonu). 

VLOOKUP ile karşılaştırınca üçüncü ve dördüncü parametrelerin olmadığını görüyoruz. VLOOKUP’ın 

üçüncü parametresi hangi kolonun değerini getirmek istediğimizi belirtirken, LOOKUP’ta hep son 

kolonun değeri gelir. VLOOKUP’ın dördüncü parametresi “Tam eşleşme / Yaklaşık eşleşme” seçimi 

sunarken, LOOKUP her zaman sıralı dizinde “Yaklaşık eşleşme” seçeneği ile çalışır. 

Vektör formu (Vector form) : 

LOOKUP(Aradığımız değer, Taranan vektör , Sonuç vektör ) 

Aradığımız değer : Numerik veya metin bir değer olabilir.  

Taranan vektör : Aranan değerin bulunduğu bir kolon veya bir satırdır (artan sırada olmak 

zorundadır). 

Sonuç vektör : Sonuç değerin bulunduğu bir kolon veya bir satırdır. 

 

 

Vektör form bize bir esneklik sağlar. Taranan vektörle sonuç vektörü farklı parametreler olduğu için, 

sağdaki vektörü (C kolonu) tarayıp, soldaki vektörden (A kolonu) sonuç getirebiliriz.  

 

Tekrar altını çizmekte fayda var, LOOKUP’ın doğru çalışabilmesi için, arama yapılan kolondaki bilgiler 

(C kolonu) artan sırada dizili olmalıdır. Sıralı olmayan listede soldaki kolonu alabilmek için INDEX - 

MATCH ikilisi kullanılır (14. örnek). 

 

Bu kadar ahde vefa yeter sanırım. LOOKUP fonksiyonunun yerini VLOOKUP, HLOOKUP, INDEX-

MATCH fonksiyonları almıştır. LOOKUP eski Excel dosyalarının uyumlu çalışabilmesi için durmaktadır.  

13 INDEX - MATCH ile VLOOKUP 
 

 

Önce INDEX - MATCH (İNDİS-KAÇINCI) için söylenen şehir efsanesini yok edelim. Bazıları der ki 

“INDEX-MATCH ikilisi VLOOKUP’tan daha hızlı çalışır”. Ne yazık ki bu doğru değildir, VLOOKUP ile 

INDEX-MATCH ikilisi aynı sürede çalışırlar. Zaman alan bölüm diziyi tarama işlemi (MATCH işlemidir), 

INDEX işlemi hızlı çalışır.  

Peki INDEX-MATCH neden popüler derseniz, “VLOOKUP’tan daha esnek çalışabildiği için” derim. 

Bu örnekte ilk olarak VLOOKUP’la aynı sonucu INDEX - MATCH ikilisi ile almayı göreceğiz. 

F6 hücresindeki araç plakasını bulmak için G6 hücresine 

=INDEX( B:B , MATCH( F6 , A:A , 0 )) 

=İNDİS( B:B ; KAÇINCI( F6 ; A:A ; 0 )) 

formülünü yazmak VLOOKUP( F6, A:B, 2, 0 ) ile aynı sonucu verir. 

MATCH işlemi bu plakanın A kolonunda kaçıncı satırda yer aldığını buldu, INDEX ise B kolonundan o 

satırdaki değeri getirdi.  

14 INDEX - MATCH ile Soldaki Değer 
 

 

INDEX - MATCH ikilisinin ayrı ayrı yazılması bize kolonları seçme serbestliğini getirmekte. İlle soldaki 

kolonda arama yapıp, sağdaki bir kolondan değer getirmek zorunda değiliz. C kolonunda arama 

yapıp, A kolonundaki değeri de getirebiliriz. 

=INDEX( A:A, MATCH( F6 , C:C , 0 ) ) 

=İNDİS( A:A ; KAÇINCI( F6 ; C:C ; 0 ) ) 

=INDEX( A:A, 9 ) 

MATCH işlemi bu plakanın C kolonunda kaçıncı satırda yer aldığını buldu, INDEX ise A kolonundan o 

satırdaki tarih değerini getirdi.  

Vektörel LOOKUP ile sıralı bir listede soldaki değere ulaşmak mümkünken, INDEX - MATCH kullanarak 

sıralı olmayan bir listede soldaki değere ulaşabiliriz.  

15 INDEX - MATCH – MATCH ile 2 Boyut 
 

 

VLOOKUP ve MATCH kullanarak bir tablodaki değeri getirmiştik. INDEX - MATCH  - MATCH birlikte 

kullanarak da aynı sonucu elde edebiliriz. 

=INDEX( A1:M11 , MATCH( A15 , A1:A11 , 0 ) , MATCH( B15 , A1:M1, 0)) 

=İNDİS( A1:M11 ; KAÇINCI( A15 ; A1:A11 ; 0 ) ; KAÇINCI( B15 ; A1:M1 ; 0)) 

Bu örnekte E15 ve F15 hücreleri MATCH formüllerini göstermek için kullanılmıştır. 

=INDEX( A1:M11 , MATCH( A15, A1:A11, 0) , MATCH( B15, A1:M1, 0 ) ) 

MATCH( “34 FJ 2216”, A1:A11, 0) = 6 

MATCH( “Nisan”, A1:M1, 0 ) = 5 

=INDEX( A1:M11 , 6 , 5 ) 

16 INDEX - MATCH – MATCH ile 3 Boyut 
 

  

VLOOKUP ve MATCH kullanarak bir tablodaki değeri getirmiştik. Aynı sonucu INDEX - MATCH  - 

MATCH ile de elde ettik. Peki tablomuzda iki değil 3 veya daha fazla değeri aramamız gerekiyorsa ne 

yapabiliriz?  

Yardımcı kolon kullanmadan 3 veya daha fazla değeri aramanın yolu dizi formülü (Array formula) 

kullanmaktır. İki kolona VLOOKUP ikinci örneğinde gördüğümüz Ctrl + Shift + Enter yöntemini bu 

örnekte de uygulayabiliriz. 

{=INDEX( A4:J13 , MATCH( A17 & "|" & B17 , A4:A13  & "|" &  B4:B13 , 0) ,  MATCH( C17 , A3:J3 , 0 ))} 

{=İNDİS( A4:J13 ; KAÇINCI( A17 & "|" & B17 ; A4:A13  & "|" &  B4:B13 ; 0) ;  KAÇINCI( C17 ; A3:J3 ; 0))} 

İlk MATCH fonksiyonu aradığımız Kod 1 | Kod 2 kombinasyonunun hangi satırda yer aldığını 

bulmakta, ikinci MATCH fonksiyonu Kod 3’ün kolonunu bulmaktadır. Bu formülü geliştirerek n sayıda 

kod kombinasyonu ile arama yapabilirsiniz. 

17 Liste VLOOKUP - COUNTIF 
 

 

VLOOKUP ile bir listeden aradığımız koşula uyan ilk değerleri getirebildiğimizi gördük. İkinci, üçüncü, 

dördüncü değerleri getirmek için “Yardımcı Kolon” yöntemini kullanabiliriz. 

 

Yukarıda B3 hücresinde “Ürün grubunu” seçtiğimizde, alttaki listenin o gruba ait tüm ürünleri 

listelemesi için “Data Filter” veya “Pivot Table” kullanabilirdik fakat biz bu işi formül ile başarmak 

istiyoruz. 

Veri alanına yardımcı bir kolon ekleyerek işe başlıyoruz. Veri sayfasında A kolonuna  

= B2 & "|" & COUNTIF ( $B$2:B2 , B2 ) 

Formülünü yazarak o ürün grubunun kaçıncı kez tekrarlandığını bulabiliriz. 

 

Liste sayfasının B3 hücresine nr_UrunGrup adını verin, veri doğrulama listesi olarak Q2:Q10 alanını 

seçin. Liste sayfasında A7:A21 alanına sıra numaralarını 1’den 15’e kadar yazın. 

Liste sayfasında B7:N21 alanına 

=IFERROR( VLOOKUP( Ürün Grubu & "|" & Satır No , Veri!$A:$O, Kolon No , 0 ) , "" )  

Formülünü yazarak listeyi oluşturun. 

=IFERROR( VLOOKUP( nr_UrunGrup & "|" & $A7; Veri!$A:$O; B$5;0);"") 

=EĞERHATA( DÜŞEYARA( nr_UrunGrup & "|" & $A7; Veri!$A:$O; B$5;0);"") 

Ben bu örneğe “VLOOKUP’ın Zirve Noktası” demeyi seviyorum. 

 

Verileri listelemenin en düz, en sade yolu yardımcı kolon kullanmaktır. 

Birazdan iki yöntem daha göreceğiz. 

18 Liste INDEX - MATCH - OFFSET 
 

 

Veri sayfasına “Yardımcı Kolon” eklemek istemiyorsanız, Liste sayfasına “Yardımcı Kolon” ekleyerek 

aynı sonuca ulaşabilirsiniz. Bu sefer formüller biraz daha uzayacak demektir. 

A6 hücresine “1” yazarak listenin birinci satırdan başlamasını garantiledikten sonra, A7:A21 alanına 

=IF(IFNUMBER(A6), IFERROR( MATCH( B3, OFFSET( Veri!$A$1 , A6 , 0 , 65000 , 1 ) , 0 ) + A6, ""), "") 

=EĞER( EĞERSAYI(A6) ; EĞERHATA( KAÇINCI( B3 ; KAYDIR( Veri!$A$1 ; A6 ; 0 ; 65000 ; 1 ) ; 0 ) + A6, 

""); "") 

Formülünü 2 yazarak hangi satırlarda o ürün grubunun yer aldığını buluruz.  

B7:N21 alanına 

=IFERROR( INDEX( Veri!B:B , $A7 ), "") 

Formülünü yazarak A kolonunda bir sayı varsa, Veri sayfasında o satırda yer alan değeri getirmesini 

sağlarız. 

                                                           
2 OFFSET( Veri!$A$1 , A6 , 0 , 65000 , 1 )    

OFFSET’in dördüncü parametresi verilen adresin uzunluğunu belirlemekte, birinci satırdan itibaren 

65000 adet satır seçmek için kullanılmaktadır. 

A6 = 1 ise, 1:65001 

A6 = 10 ise 10:65009 satırları taranır. 

Tam hesaplama yapmak için OFFSET( Veri!$A$1 , A6 , 0 , 65000 - A6, 1 )  formülü kullanılabilirdi. 

( Yardımcı kolon bulunmamakta ) 

INDEX - MATCH ikilisinin hızlı çalıştığı durum bu örnekte gerçekleşmektedir. Zaman alan MATCH 

işlemi bir kere yapılmakta ( Liste isimli çalışma sayfasının A kolonunda ), B, C, D, ...  kolonlarında 

INDEX kullanıldığı için veriler daha hızlı gelmektedir. 

 

Verileri listelemenin en performanslı yöntemi INDEX - MATCH - OFFSET kullanmaktır. 

INDEX - MATCH ikilisinin hızlı çalışmasıyla ilgili şehir efsanesi, her satırda MATCH’in sadece bir kez, 

INDEX’in birden çok defa kullanıldığı bu yönteme dayanmaktadır. 

19 Liste Ctrl - Shift - Enter 
 

 

Bundan önceki iki örnekte “Yardımcı Kolon” kullanarak seçtiğimiz ürün grubunu listeledik. Bu örnekte 

dizi formülü (Array Formula) kullanarak “Yardımcı Kolon”dan tamamen kurtulacağız. 

İlk olarak bazı ad tanımlamaları yapalım. Veri sayfasında listenin başlık satırını “nrBaslik” olarak, 

listenin tamamını “nrListe”,  ürün adlarının yer aldığı A kolonunu “nrUrunGrupListe” olarak 

tanımlayın. 

 

Liste sayfasında B3 hücresine nrUrunGrup ismini verin. 

B7 hücresine aşağıdaki formülü yazıp Ctrl + Shift + Enter tuşlarına basın. 

{=IFERROR(  INDEX( nrListe ; SMALL( IF( nrUrunGrupListe=nrUrunGrup ; ROW( nrUrunGrupListe ) ) ; 

$A7 ) - ROW( nrBaslik ) ; COLUMN( B$1 ) ) ; "" )} 

{=EĞERHATA(  İNDİS( nrListe ; KÜÇÜK( EĞER( nrUrunGrupListe=nrUrunGrup ; SATIR( nrUrunGrupListe 

) ) ; $A7 ) - SATIR( nrBaslik ) ; SÜTUN( B$1 ) ) ; "" )} 

Bu formülü B7:N21 alanına kopyalayın. 

 

Formülün açılımı : 

Formülün “IF( nrUrunGrupListe=nrUrunGrup , ROW( nrUrunGrupListe ) )” kısmını seçip F9 3 tuşuna 

basınca : 

{FALSE\FALSE\FALSE\FALSE\FALSE\7\8\9\10\11\12\13\14\15\16\17\18\FALSE\FALSE\FALSE\FALSE\

FALSE\FALSE\FALSE\FALSE\FALSE\FALSE\FALSE\FALSE\FALSE\FALSE\FALSE\FALSE\FALSE\FALSE\FALS

E\FALSE\FALSE\FALSE\FALSE\FALSE\FALSE\FALSE\FALSE\FALSE\FALSE\FALSE\FALSE\FALSE\FALSE\FA

LSE\FALSE\FALSE\FALSE\FALSE\FALSE\FALSE\FALSE\FALSE\FALSE\FALSE\FALSE\FALSE\FALSE\FALSE\

FALSE\FALSE\FALSE\FALSE\FALSE\FALSE\FALSE\FALSE} 

değerini verdiğini görürsünüz. 

“SMALL( IF( nrUrunGrupListe=nrUrunGrup ; ROW( nrUrunGrupListe ) ) ; $A7 )” A7 = 1 olduğu için en 

küçük birinci değeri, 7 değerini getirir.  

“INDEX( nrListe ; SMALL( IF( nrUrunGrupListe=nrUrunGrup ; ROW( nrUrunGrupListe ) ) ; $A7 ) - ROW( 

nrBaslik ) ; COLUMN( B$1 ) )” 7. satırdaki değerleri kolon kolon getirir. 

Bir sonraki satırda $A8 = 2 olduğu için 8. satırdaki değerler gelir, daha sonra 9, 10, 11, … 18’e kadar 

gider. 

 

Verileri listelemeyi üç farklı yöntemle gördük.  

Bu yöntemde hiç yardımcı kolon kullanmadık, tek bir formül ile sonuca ulaştık.  

  

                                                           
3 Formül içindeyken F9 tuşuna basınca, formülün seçilen alanını (veya tamamını) hesaplayıp değerini 
getirmekte. İçiçe karışık formülleri izlemek için oldukça faydalı bir kısayol.  
F9’un geri dönüşü olmadığı için seçili alanı önce Ctrl-C ile kopyalayıp, sonra F9’a basmak, formüle geri dönmek 
için Ctrl-V’ye basmak aklınızda olsun. 

20 VLOOKUP 3 Kolon Ctrl - Shift - Enter 
 

 

Yardımcı kolon kullanmadan iki kolona VLOOKUP yapmayı 7 numaralı örnekte görmüştük (Ctrl-Shft-

Enter ile). İkiden fazla kolon için aynı formülü devam ettirebileceğimizi de not etmiştik. 

Bu örnekte yardımcı kolon kullanmadan 3 kolona VLOOKUP yapacağız. 

J2 hücresine 

=VLOOKUP(G3 & "|" & H3 & "|" & I3, CHOOSE({1,2},A:A  & "|" &B:B  & "|" &C:C,D:D ), 2, 0) 

=DÜŞEYARA(G3 & "|" & H3 & "|" & I3 ; ELEMAN({1,2},A:A  & "|" &B:B  & "|" &C:C ; D:D ); 2; 0) 

Formülünü yazıp “Ctrl + Shift + Enter” tuşlarına birlikte basın. Böylece formülünüz 

{=VLOOKUP(G3 & "|" & H3 & "|" & I3, CHOOSE({1,2},A:A  & "|" &B:B  & "|" &C:C,D:D ), 2, 0)} 

şeklini alacaktır. 

 

Ctrl-Shft-Enter yönteminin performansı düşürdüğünü gördüğüm için kullanmaktan hep kaçındım. 

Çok sayıda Ctrl-Shft-Enter formülü kullanınca, Excel dosyada çalışmak uzun zaman almaya başlıyor.  

Bu örneği yardımcı kolon kullanan yöntemle değiştirerek ciddi performans artışı sağlayabilirsiniz. 

21 VLOOKUP 5 Farklı Tablodan 
 

 

2013 fiyatları bir sayfada, 2012, 2011, 2010 ve 2009 fiyatları ise ayrı sayfalarda. B kolonunda seçilen 

yıla göre farklı tabloya bakmak için “INDIRECT” formülünü kullanabilirsiniz. 

=VLOOKUP( $A2 , INDIRECT( $B2 & "!A:D" ) , 2 , 0 ) 

=VLOOKUP( $A2 , 2013!A:D , 2 , 0 ) 

=DÜŞEYARA( $A2 ; DOLAYLI( $B2 & "!A:D" ) ; 2 ; 0 ) 

 

INDIRECT( $B2 & "!A:D" ) formülü 4 ile B2 hücresine yazılan çalışma sayfası ismi (worksheet name) 

formüllerde kullanılmaktadır.  

INDIRECT formülü ile verilen metin referans, alan adresi olarak kullanılır. B2 hücresindeki 2013 

değeri “!A:D” ile birleştirilerek 2013!A:D alanındaki değerler taranır ve sonuç alınır. B3 hücresinde 

2012 değeri olduğu için 2012!A:D alanındaki değerler taranır. 

 

  

                                                           
4 Çalışma sayfası isminde boşluk varsa INDIRECT("'" & $B2 & "'!A:D" ) şeklinde formül kullanılır. 



41 
 

22 SUMIFS ile Toplam 
 

 

VLOOKUP ile numerik/alfanumerik ilk değerleri getirebilirsiniz. Eğer koşula uyan tüm değerleri 

toplamanız gerekirse SUMIFS (ÇOKETOPLA) işinizi kolayca görecektir. 

=SUMIFS ( C:C , A:A , E9 , B:B , F9 ) 

=ÇOKETOPLA( C:C ; A:A ; E9 ; B:B ; F9 ) 

Sonsöz  
 

Yukarıdaki örneklerle VLOOKUP’ın sınırlarını aşmayı gördünüz. Bu yöntemleri saklayın, yeri 

geldiğinde, ihtiyacınız olduğunda, sakladığınız yerden çıkartıp kullanın. Excel bilginizi çalışma 

hayatınızda avantaja dönüştürün. 

Başta da söylediğim gibi yüzbinlerce satır veriyle uğraşıyorsanız Excel’den daha uygun çözümler 

mevcuttur. Her ay tekrarlanan konularda, fazla mesai harcayarak, sistem limitlerini zorlayarak 

raporlar hazırlıyorsanız artık bilgi işlem bölümünden (veya dış kaynaklardan) destek alma zamanı 

gelmiş demektir.  

Tüm görüşlerinizi excel@vegatr.com adresine gönderebilirsiniz. 

 

 

  

mailto:excel@vegatr.com

